
Mix.install([
  {:unpickler, "~> 0.1.0"},
  {:nx, "~> 0.4.0"}
])
:ok
Setup
We will first define a helper macro to automatically generate a pickle binary from example Python code.
defmodule Pickle do
  @doc """
  Runs the given Python code and returns the pickle binary.

  The code must define a variable `x` and that value is pickled.

  Optionally, a numeric sigil modifier may be given indicating
  the pickle version to use, defaults to version 4.

  Note that a `python` executable must be available.

  ## Examples

      iex> import Pickle
      iex> ~P'''
      ...> x = 1
      ...> '''
      <<128, 4, 75, 1, 46>>

  """
  defmacro sigil_P({:<<>>, _meta, [code]}, modifiers) do
    protocol =
      case modifiers do
        [] -> 4
        [version] -> version - ?0
      end

    code = """
    import pickle
    import base64

    #{code}

    data = pickle.dumps(x, protocol=#{protocol})
    encoded = base64.b32encode(data).decode("utf-8")
    print(encoded, end="")
    """

    {encoded, 0} = System.cmd("python", ["-c", code])
    binary = Base.decode32!(encoded)
    Macro.escape(binary)
  end
end
1 doctest, 0 failures
{:module, Pickle, <<70, 79, 82, 49, 0, 0, 11, ...>>, {:sigil_P, 2}}
import Pickle
Pickle
Basic data structures
~P"""
x = 1
"""
|> Unpickler.load!()
{1, ""}
~P"""
x = [1, 2, 3]
"""
|> Unpickler.load!()
{[1, 2, 3], ""}
~P"""
x = [1, 2.0, "text", (None, True), {"key": "val"}, b"\x01\x00"]
"""
|> Unpickler.load!()
{[1, 2.0, "text", {nil, true}, %{"key" => "val"}, <<1, 0>>], ""}
Custom object resolver
All non-trivial objects are deserialized as %Unpickler.Object{} structs. For example let's consider a date object:
~P"""
from datetime import date
x = date.fromisoformat("2022-05-17")
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "datetime.date",
   args: [<<7, 230, 5, 17>>],
   kwargs: %{},
   state: nil,
   append_items: [],
   set_items: []
 }, ""}
However, ideally we would want to load that object as Elixir %Date{}. To do that, we can specify an object resolver:
object_resolver = fn
  # See https://github.com/python/cpython/blob/3.10/Lib/datetime.py#L1094-L1105
  %Unpickler.Object{constructor: "datetime.date", args: [<<year_hi, year_lo, month, day>>]} ->
    {:ok, date} = Date.new(year_hi * 256 + year_lo, month, day)
    {:ok, date}

  _ ->
    :error
end

~P"""
from datetime import date
x = date.fromisoformat("2022-05-17")
"""
|> Unpickler.load!(object_resolver: object_resolver)
{~D[2022-05-17], ""}
Note that %Unpickler.Object{} has a number of different fields, that's because Python objects may define multiple ways of serializing themselves and we gather all the information under a single unified struct. This way we can easily pattern match on it in the object resolver.
When deserializing complex objects you generally need to look at the corresponding Python source code and reverse-engineer how to load them accordingly. Here are a couple examples:
~P"""
class Point:
  def __init__(self, x, y):
    self.x = x
    self.y = y

x = Point(1, 1)
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "__main__.Point.__new__",
   args: [%Unpickler.Global{scope: "__main__", name: "Point"}],
   kwargs: %{},
   state: %{"x" => 1, "y" => 1},
   append_items: [],
   set_items: []
 }, ""}
~P"""
class Point:
  def __init__(self, x, y):
    self.x = x
    self.y = y

  def __reduce__(self):
    return (Point, (self.x, self.y))

x = Point(1, 1)
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "__main__.Point",
   args: [1, 1],
   kwargs: %{},
   state: nil,
   append_items: [],
   set_items: []
 }, ""}
~P"""
class Point:
  def __init__(self, x, y):
    self.x = x
    self.y = y

  @classmethod
  def _reconstruct(cls, x, y):
    return cls(x, y)

  def __reduce__(self):
    return (Point._reconstruct, (self.x, self.y))

x = Point(1, 1)
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "__main__.Point._reconstruct",
   args: [1, 1],
   kwargs: %{},
   state: nil,
   append_items: [],
   set_items: []
 }, ""}
~P"""
class Point:
  def __init__(self, x, y):
    self.x = x
    self.y = y

  def __getstate__(self):
    return (self.x, self.y)

  def __setstate__(self, state):
    (x, y) = state
    self.x = x
    self.y = y

x = Point(1, 1)
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "__main__.Point.__new__",
   args: [%Unpickler.Global{scope: "__main__", name: "Point"}],
   kwargs: %{},
   state: {1, 1},
   append_items: [],
   set_items: []
 }, ""}
~P"""
class Point:
  def __init__(self, x, y):
    self.x = x
    self.y = y

  def __reduce__(self):
    append_iter = iter([1, 2])
    set_iter = iter([("a", 1), ("b", 2)])
    return (Point, (self.x, self.y), None, append_iter, set_iter)

x = Point(1, 1)
"""
|> Unpickler.load!()
{%Unpickler.Object{
   constructor: "__main__.Point",
   args: [1, 1],
   kwargs: %{},
   state: nil,
   append_items: [1, 2],
   set_items: [{"a", 1}, {"b", 2}]
 }, ""}

Unpickler is a library for loading data in the Python's
pickle format.
Supports all pickle protocols from 0 to 5.

          @spec load!(
  binary(),
  keyword()
) :: {term(), rest :: binary()}


      


Loads the given pickle binary.
Basic literals and data structures are deserialized as corresponding
Elixir terms whenever possible. The pickle data may include arbitrary
Python objects, so in all other cases the Unpickler.Object struct
is used. This struct holds all the information that would be used for
object reconstruction. You can define a custom :object_resolver
function to recognise certain objects and map them to whatever data
structure you see fit.

  
  object-references

  
  Object references


Note that if the object hierarchy includes circular references, it
is inherently impossible to represent in Elixir. One example of such
data structure is a list with a reference to itself:
x = []
x.append(x)
On the other hand, multiple references to the same object are restored
as expected, without duplicating memory, as in:
x = [1, 2, 3]
y = (x, x)

  
  options

  
  Options


	:object_resolver - a function for constructing a custom term
corresponding to a Python object. Receives Unpickler.Object
as an argument and should return the term as {:ok, term} or
:error if not applicable

	:persistent_id_resolver - a function returning an object for
the given persistent id. This function is required if the data
includes a persistent id, otherwise an error is raised



  
  examples

  
  Examples


The scalar 1 would be loaded like so:
iex> Unpickler.load!(<<128, 4, 75, 1, 46>>)
{1, ""}
Next, a more complex data structure:
[1, 2.0, "text", (None, True), {"key": "val"}, b"\x01\x00"]
iex> data =
...>   <<128, 4, 149, 47, 0, 0, 0, 0, 0, 0, 0, 93, 148, 40, 75, 1, 71, 64, 0, 0, 0, 0, 0, 0, 0,
...>     140, 4, 116, 101, 120, 116, 148, 78, 136, 134, 148, 125, 148, 140, 3, 107, 101, 121,
...>     148, 140, 3, 118, 97, 108, 148, 115, 67, 2, 1, 0, 148, 101, 46>>
iex> Unpickler.load!(data)
{[1, 2.0, "text", {nil, true}, %{"key" => "val"}, <<1, 0>>], ""}

  
  objects

  
  Objects


Other objects end up as Unpickler.Object
from datetime import date
date.fromisoformat("2022-05-17")
iex> data =
...>   <<128, 4, 149, 32, 0, 0, 0, 0, 0, 0, 0, 140, 8, 100, 97, 116, 101, 116, 105, 109, 101,
...>     148, 140, 4, 100, 97, 116, 101, 148, 147, 148, 67, 4, 7, 230, 5, 17, 148, 133, 148, 82,
...>     148, 46>>
iex> Unpickler.load!(data)
{%Unpickler.Object{
   append_items: [],
   args: [<<7, 230, 5, 17>>],
   constructor: "datetime.date",
   kwargs: %{},
   set_items: [],
   state: nil
 }, ""}
For those, we can customize construction by specifying an :object_resolver
from datetime import date
date.fromisoformat("2022-05-17")
iex> data =
...>   <<128, 4, 149, 32, 0, 0, 0, 0, 0, 0, 0, 140, 8, 100, 97, 116, 101, 116, 105, 109, 101,
...>     148, 140, 4, 100, 97, 116, 101, 148, 147, 148, 67, 4, 7, 230, 5, 17, 148, 133, 148, 82,
...>     148, 46>>
iex> object_resolver = fn
...>   # See https://github.com/python/cpython/blob/3.10/Lib/datetime.py#L1094-L1105
...>   %{constructor: "datetime.date", args: [<<year_hi, year_lo, month, day>>]} ->
...>     {:ok, date} = Date.new(year_hi * 256 + year_lo, month, day)
...>     {:ok, date}
...>
...>   _ ->
...>     :error
...> end
iex> Unpickler.load!(data, object_resolver: object_resolver)
{~D[2022-05-17], ""}

  


        

      



  

    
Unpickler.Global 
    



      
Represents a reference to a global class or function.

      





  

    
Unpickler.Object 
    



      
Represents a Python object.
This struct includes all the information that would be used for
object reconstruction. Refer to the pickle documentation for more
details.
Information
The following fields are available:
	:constructor - name of the class or function called to build
the object

	:args - a list of arguments passed to the constructor

	:kwargs - a map with keyword arguments passed to the constructor

	:state - a value passed to the __setstate__ method if applicable

	:append_items - a list of values to append to the object

	:set_items - a list of key-value pairs to set on the object

